

Vorgespannte Substrukturen mit ANSYS

Ausgabe: 05 / 2005

Problem:


Substrukturen werden in ANSYS eingesetzt, um bei großen Modellen die Rechenzeit zu verkürzen. Bei manchen Analysen ist es wünschenswert, einen Vorspannungszustand zu berücksichtigen, um die durch eine statische Vorlast (z.B. Eigengewicht) entstehende Spannungsversteifung mit abzubilden. Solche Analysen sind in ANSYS möglich, ohne daß vor der Substrukturanalyse die statische Vorbelastung berechnet werden muß.

Erläuterung:

Die Berechnung wird in mehreren Schritten durchgeführt. Zuerst werden in einem Substruktur – generation pass konventionelle Substrukturen erzeugt und im nachfolgenden use pass die statische Vorbelastung aufgebracht. Anschließend wird ein expansion pass für die Substruktur(en) durchgeführt mit dem Kommando `pstres,on`. Dadurch werden die spannungsversteifenden Effekte mitexpandiert. Anschließend wird der generation pass für die Substruktur(en) ein zweites Mal durchgeführt und Platz für die prestress Matrix gelassen (`seopt,name,2,0,1`). Im zweiten use pass wird nun die vorgespannte Analyse durchgeführt (statisch, modal oder harmonic). Ein zweiter Expansionpass liefert die Ergebnisse der vorgespannten Analyse.

Beispiel:

Das beiliegende Beispiel zeigt den Workflow anhand einer einfachen Geometrie. Zum Vergleich wurde das Modell konventionell und als (expandierte) Substrukturanalyse mit Vorspannung gerechnet.



KLS

Vorgespannte Substrukturen mit ANSYS

Ausgabe: 05 / 2005

ANSYS Eingabesatz (ANSYS 9.0):

**Generation pass
konventionelle Superelemente**

**Use pass
mit statischer Vorbelastung**

**Expansion pass
mit pstres,on**

**2. Generation pass
Platz lassen für Prestress Matrix**

**2. Use pass
verwendet die Vorspannung**

**2. Expansion pass
verwendet die Vorspannung**

```
fini
/out,out
/clear,nosta
/out
/filename,full $ /prep7
/title,Superelement with Omega(100r/s)
et,1,shell63 $ r,1,0.5
mp,ex,1,1e7 $ mp,dens,1,0.001
csys,1
/com,geometrie
k,1,10 $ k,2,20 $ k,3,20,45
k,4,10,45 $ a,1,2,3,4
esize,,2 $ amesh,1 $ agen,8,1,,,45
nummrg,all
csys,1 $ nsel,s,loc,x,10 $ nrotate,all
alls $ save $ fini
```

```
*do,iii,1,8
/com generate Substructure %iii%
/filename,gen%iii%
/solu
antype,substr
asel,s,,,iii
esla $ nsel $ m,all,all
seopt,gen%iii%,2,0,1
omega,,,100.0
solve
fini
*enddo
alls
/filn,full
save
fini
/clear
/com,generate model with 8
Substructures
/filename,use
/prep7
et,1,50
*do,iii,1,8
se,gen%iii%
*enddo
alls
csys,1
nsel,s,loc,x,10 $ d,all,uy $ d,all,uz
alls
save
fini
/com,first use pass
/solu
antype,static
sfe,all,1,selv,,1.0
solve
fini
save,use,db
resume,full,db
/com,perform 8 expansion passes
*do,iii,1,8
/filename,gen%iii%
/solu
expass,on
seexp,gen%iii%,use
pstres,on
expsol,1,1,,yes
solve
fini
*enddo
```

Vorgespannte Substrukturen mit ANSYS

Ausgabe: 05 / 2005

ANSYS Eingabesatz (ANSYS 9.0):

```
/filn,full
resu
/com,Now do stress-stiffened gen pass
*do,iii,1,8
  /filename,gen%iii%
  /solu
  antype,substr
  pstres,on
  seopt,gen%iii%,2,0,1
  solve
  fini
*enddo
fini $ /clear

/filename,modal
/prep7
et,1,50
*do,iii,1,8
  se,gen%iii%
*enddo
alls $ save $ fini

/com,Now do stress-stiffened use pass
/solu
antype,modal
modopt,lanb,10
sfe,all,1,selv,,1.0
mxpand,10
solve
```

```
fini $ /clear

/filn,full
resu
*do,iii,1,8
  /filename,gen%iii%
  /solu
  antype,substr
  expass,on
  seexp,gen%iii%,modal
  expsol,1,7,,yes
  solve
  fini
*enddo
fini

/filn,full
resu
/post1
*do,iii,1,8
  file,gen%iii%
  set,last
*enddo
/post1
/view,1,1,1,1
plns,u,sum,2
```

Die Input Datei können Sie unter folgender Adresse herunterladen:

http://www.cadfem.de/fileadmin/files/9_service_newsletter/2005/0505/pssub.inp